

## Possible Permutation Characters

One can get candidates for permutation characters by testing linear combinations of irreducible characters with nonnegative integer coefficients for fulfilling certain necessary conditions. While there is no guarantee, usually most of the resulting characters are bona fide permutation characters.

```
gap> c:=CharacterTable("A5");
CharacterTable( "A5" )
gap> p:=PermChars(c);
[ Character( CharacterTable( "A5" ), [ 1, 1, 1, 1, 1 ] ),
  Character( CharacterTable( "A5" ), [ 5, 1, 2, 0, 0 ] ),
  Character( CharacterTable( "A5" ), [ 6, 2, 0, 1, 1 ] ),
  Character( CharacterTable( "A5" ), [ 10, 2, 1, 0, 0 ] ),
  Character( CharacterTable( "A5" ), [ 12, 0, 0, 2, 2 ] ),
  Character( CharacterTable( "A5" ), [ 15, 3, 0, 0, 0 ] ),
  Character( CharacterTable( "A5" ), [ 20, 0, 2, 0, 0 ] ),
  Character( CharacterTable( "A5" ), [ 30, 2, 0, 0, 0 ] ),
  Character( CharacterTable( "A5" ), [ 60, 0, 0, 0, 0 ] ) ]
```

Sylow theorems show easily that in this case all must be proper permutation characters.

Note that  $\pi_2 = \chi_1 + \chi_4$  and  $\pi_4 = \chi_1 + \chi_4 + \chi_5$ , while the corresponding subgroups are not contained in each other, thus subgroup inclusion cannot be determined from permutation characters.

## Determining subgroups from a character table

We want to study subgroups of  $M_{11}$ , based on information in its character table. We start by obtaining the table and the potential permutation characters:

```
gap> c:=CharacterTable("M11");;
gap> p:=PermChars(c);
[ Character( CharacterTable( "M11" ), [ 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1 ] ),
  Character( CharacterTable( "M11" ), [ 11, 3, 2, 3, 1, 0, 1, 1, 0, 0 ] ),
  Character( CharacterTable( "M11" ), [ 12, 4, 3, 0, 2, 1, 0, 0, 1, 1 ] ),
  [...]
  Character( CharacterTable( "M11" ), [ 7920, 0, 0, 0, 0, 0, 0, 0, 0, 0 ] ) ]
```

Lets start looking at the 2-Sylow subgroup. As  $7920 = 2^4 3^2 5 \cdot 11$  it has index  $7920/16 = 495$ . There is only one potential permutation character of this degree.

```
gap> pc:=Filtered(p,i->i[1]=495);
[ Character( CharacterTable( "M11" ), [ 495, 15, 0, 3, 0, 0, 1, 1, 0, 0 ] ) ]
gap> pc:=pc[1];
```

Let  $U$  be a 2-Sylow subgroup. From the formula for the induced class function, we can calculate how many elements of  $U$  lie in each class of  $G$ :

```
gap> List([1..10], i->pc[i]/pc[1]*SizesConjugacyClasses(c)[i]);
[ 1, 5, 0, 6, 0, 0, 2, 2, 0, 0 ]
gap> OrdersClassRepresentatives(c);
[ 1, 2, 3, 4, 5, 6, 8, 8, 11, 11 ]
```

We have one element of order 1, five elements of order 2, six elements of order 4 and twice two elements of order 8. Inspecting the list of groups of order 16, this already uniquely identifies the isomorphism type of  $U$ .

```
gap> NrSmallGroups(16);
14
gap> List([1..14], i->Collected(List(Elements(SmallGroup(16,i)), Order)));
[ [ [ 1, 1 ], [ 2, 1 ], [ 4, 2 ], [ 8, 4 ], [ 16, 8 ] ], [...]
  [ [ 1, 1 ], [ 2, 5 ], [ 4, 6 ], [ 8, 4 ] ], [...] ]
```

Lets now look at the subgroup  $S$  of index 11 (a point stabilizer) in more detail. Again we get numbers of elements as well as the indices of the classes in which the elements lie: 45 order 2; 80 order 3, etc.

```
gap> pc:=p[2];
Character( CharacterTable( "M11" ), [ 11, 3, 2, 3, 1, 0, 1, 1, 0, 0 ] )
gap> sz:=List([1..10], i->pc[i]/pc[1]*SizesConjugacyClasses(c)[i]);;
gap> sz;OrdersClassRepresentatives(c);
[ 1, 45, 80, 270, 144, 0, 90, 90, 0, 0 ]
[ 1, 2, 3, 4, 5, 6, 8, 8, 11, 11 ]
```

Note that these element order do not agree with  $S_6$ , which has the same order, the subgroup thus is not isomorphic to  $S_6$  (or similarly  $A_6 \times 2$ ).

```
gap> Collected(List(Elements(SymmetricGroup(6)), Order)); # inefficient
[ [ 1, 1 ], [ 2, 75 ], [ 3, 80 ], [ 4, 180 ], [ 5, 144 ], [ 6, 240 ] ]
```

Now let us assume that this subgroup  $S$  has a subgroup  $T$  of index 2, i.e. index 22 in the group. There is only one possible permutation character:

```
gap> qc:=Filtered(p, i->i[1]=22);
[ Character( CharacterTable( "M11" ), [ 22, 6, 4, 2, 2, 0, 0, 0, 0, 0 ] ) ]
gap> qc:=qc[1];;
```

We again count how many elements of what order are there, and observe that this agrees with the element count in  $A_6$ :

```
gap> tz:=List([1..10], i->qc[i]/qc[1]*SizesConjugacyClasses(c)[i]);;
gap> tz;OrdersClassRepresentatives(c);
[ 1, 45, 80, 90, 144, 0, 0, 0, 0, 0 ]
[ 1, 2, 3, 4, 5, 6, 8, 8, 11, 11 ]
gap> Collected(List(Elements(AlternatingGroup(6)), Order)); # inefficient
[ [ 1, 1 ], [ 2, 45 ], [ 3, 80 ], [ 4, 90 ], [ 5, 144 ] ]
```

Can we do more? We try to calculate the character table of this subgroup. Note that these element counts already allow us to calculate the inner product of reduced characters:

Based on these numbers, we can already calculate the scalar product of restrictions of characters of  $M_{11}$  to  $S$ , by combining the classes that fuse the same.

```
gap> sel:=Filtered([1..10],i->sz[i]>0);
[ 1, 2, 3, 4, 5 ]
gap> prdrest:=function(a,b)
> return qc[1]/Size(c)*Sum(sel,i->tz[i]*a[i]*GaloisCyc(b[i],-1));end;
gap> gram:=List(Irr(c),i->List(Irr(c),j->prdrest(i,j)));;
gap> Display(gram);
[ [ 1, 1, 0, 0, 1, 0, 0, 0, 0, 0 ],
  [ 1, 2, 0, 0, 1, 0, 0, 2, 1, 1 ],
  [ 0, 0, 1, 1, 0, 0, 0, 0, 2, 2 ],
  [ 0, 0, 1, 1, 0, 0, 0, 0, 2, 2 ],
  [ 1, 1, 0, 0, 3, 0, 0, 2, 0, 2 ],
  [ 0, 0, 0, 0, 0, 2, 2, 2, 2, 2 ],
  [ 0, 0, 0, 0, 0, 2, 2, 2, 2, 2 ],
  [ 0, 2, 0, 0, 2, 2, 2, 8, 4, 6 ],
  [ 0, 1, 2, 2, 0, 2, 2, 4, 7, 7 ],
  [ 0, 1, 2, 2, 2, 2, 2, 6, 7, 9 ] ]
```

This tells us that  $\chi_1$ ,  $\chi_2 - \chi_1$  and  $\chi_3$ , ( $\chi_4$  reduces the same) restrict to irreducible characters. We also take the restrictions of the remaining characters.

```
gap> s:=[Irr(c)[1],Irr(c)[2]-Irr(c)[1],Irr(c)[3],Irr(c)[4],Irr(c)[6]];;
gap> r:=Irr(c){[5,6,8,9,10]}; # 7 reduces same as 6
```

Next we reduce the restricted characters with the restricted irreducibles. We note that two of the reductions are the same and discard.

Two reduce completely, two proper reducible characters, each of norm 2, remain.

```
gap> List(r,i->List(s,j->prdrest(i,j)));
[ [ 1, 0, 0 ], [ 0, 0, 0 ], [ 0, 2, 0 ], [ 0, 1, 2 ], [ 0, 1, 2 ] ]
gap> r[1]:=r[1]-s[1];;
gap> r[3]:=r[3]-2*s[2];;
gap> r[4]:=r[4]-s[2]-2*s[3];;
gap> r[5]:=r[5]-s[2]-2*s[3];;
gap> r[2]{sel}=r[4]{sel};
true
gap> r[3]{sel}=r[5]{sel};
true
gap> r:=r{[1..3]};;
```

Next we form mutual inner products. This tells us that the (reduction) of the third character must be a sum of the reductions of the first two.

```
gap> List(r,i->List(r,j->prdest(i,j)));
[ [ 2, 0, 2 ], [ 0, 2, 2 ], [ 2, 2, 4 ] ]
gap> (r[1]+r[2]){sel}=r[3]{sel};
true
gap> r:=r{[1,2]};
gap> List(r,i->i[1]);
[ 10, 16 ]
```

These remaining two reducible characters have degrees 10 and 16. The inner products tell us that they must have the form  $\chi_1 + \chi_2$  and  $\chi_3 + \chi_4$  where all four characters are irreducible characters of  $S$  that are not yet known. And since every irreducible character must occur in a reduction, there are just four extra irreducible characters (thus 7 in total).

```
gap> List(s,x->x[1]);
[ 1, 9, 10 ]
gap> 360-(1+9^2+10^2);
178
gap> RootInt(178);
13
```

We now look at the list of irreducible restrictions. Their degree squares sum up to 182, so the remaining (at least four) characters must have degree squares summing up to 178. We now test for possible degrees from the numbers in  $\{1, \dots, 13\}$  so that squares sum up to 178, and we can combine both 10 and 16 as sum of two of them, using different numbers:

```
gap> len:=4;;Filtered(UnorderedTuples([1..13],len),i->Sum(i,j->j^2)=178
> and ForAny(Combinations([1..len],2),j->Sum(i{[j]}=10 and
> ForAny(Combinations(Difference([1..len],j),2),k->Sum(i{k})=16)));
[ [ 5, 5, 8, 8 ] ]
```

So the character degrees must be 1, 5, 5, 8, 8, 9, 10. We also check whether any of the unions of classes we found must split:

```
gap> List(tz{sel},i->360/i);
[ 360, 8, 9/2, 4, 5/2 ]
```

So the elements in classes 3 and 5 (also of orders 3 and 5 by happenstance) must each split up into two classes under the subgroup, which gives us the required 7 classes.

To determine class orders we note that the class orders must add up to 80, respectively 144. Class orders also are the indices of the element centralizers and thus (the elements centralize themselves) must divide  $360/3$ , respectively  $360/5$ . This tells us that the 144 elements of order 5 must be in two classes of order 72 each, while the elements of order 3 could split into two classes of order 40, or classes of order 20 and 60.

```

gap> tz{sel{[3,5]}};
[ 80, 144 ]
gap> d:=DivisorsInt(360/5);
[ 1, 2, 3, 4, 6, 8, 9, 12, 18, 24, 36, 72 ]
gap> Filtered(UnorderedTuples(d,2),x->Sum(x)=144);
[ [ 72, 72 ] ]
gap> d:=DivisorsInt(360/3);
[ 1, 2, 3, 4, 5, 6, 8, 10, 12, 15, 20, 24, 30, 40, 60, 120 ]
gap> Filtered(UnorderedTuples(d,2),x->Sum(x)=80);
[ [ 20, 60 ], [ 40, 40 ] ]

```

For the two possibilities of class orders we test for possible nontrivial normal subgroups, that must be unions of classes, including the identity

```

gap> Filtered(Combinations([45,20,60,90,72,72]),
> x->Length(x)>0 and 360 mod (1+Sum(x))=0);
[ [ 20, 45, 60, 72, 72, 90 ] ]
gap> Filtered(Combinations([45,40,40,90,72,72]),
> x->Length(x)>0 and 360 mod (1+Sum(x))=0);
[ [ 40, 40, 45, 72, 72, 90 ] ]

```

In both cases the only possibility is the whole subgroup. We thus have shown that if  $M_{II}$  has a subgroup of index 22, it must be simple (and thus must be isomorphic to  $A_6$ ).

With the same methods we can calculate the character table of the point stabilizer  $S$  and note that it has two linear characters (and thus such a subgroup in it of index 2). It thus is a group of form  $A_6.2$ , that is isomorphic to neither  $S_6$ , nor  $A_6 \times 2$ . It is in fact an extension with one of the surprising automorphisms of  $A_6$ .